

UniDyn--Demo-05.nb

John A. Marohn
jam99@cornell.edu
Cornell University

Abstract: Use the **UniDyn** Evolver function to calculate the evolution of the magnetization of a two coupled spin = 1/2 particles under free evolution, a heteronuclear COSY pulse sequence, and a homonuclear pulse sequence.

Instructions

Once you edit the directory path below, you should be able to run the notebook. This notebook will calculate, for two protons that are "J" coupled, the free induction decay and the two-dimensional CORrelation SpectroscopY (COSY) signal. Both the heteronuclear (e.g., a coupled ^{13}C and ^1H) and homonuclear (e.g., two coupled ^1H 's) experiments are written out below. I have highlighted in yellow and grey the places in the file where you may want to change parameters.

Answer these questions after lab.

1) Free evolution signal

I give some suggested values for the two proton chemical shifts, the J coupling, and the transverse relaxation time that will give nice looking spectra.

a) Edit the J-coupling to zero and re-run the free-induction decay calculation. Does the spectrum look like you expect? (Reset the J-coupling back to something reasonable before continuing.)

b) Below I plot the real part of the Fourier transform. Now plot the imaginary part and magnitude. Is the resolution better for the real part of the signal or the magnitude part? Can you explain the result qualitatively?

2) COSY. Each COSY signal takes a few seconds to calculate.

a) Draw out the pulse sequence for the Heteronuclear COSY and the Homonuclear COSY experiments. How are the experiments different?

b) In each spectrum, there is a small signal blip at the center of the plot. This is not due to the spins -- I added that fake signal. How did I add it? Why did I add it?

For both the Heteronuclear COSY and the Homonuclear COSY experiments,

c) Edit the J-coupling to zero and recalculate the spectrum. Does the COSY spectrum look any different? (Reset the J-coupling back to something reasonable before continuing.)

d) Below I plot the two-dimensional Fourier transform of the COSY signal. The signal is not symmetric in ω_1 and ω_2 . Can you explain this? Look at the terms in the ρ_4^A , ρ_4^B , and ρ_4 signals. How was each of these signals created?

e) Like you did with the FID, plot the imaginary, real, and magnitude of the two-dimensional Fourier transform of the COSY(t_1, t_2) signal -- these are the plots **p1**, **p2**, and **p3** returned by the 2D Fourier transform function. Note the peak shapes. For example, are all the peaks positive? To see whether the peaks in the spectrum are positive or negative, you can rotate the plot around using the mouse.

Preliminaries

Set the path to the package

Tell *Mathematica* the path to the directory containing the package.

EDIT THE FOLLOWING PATH STRING:

```
$PackagePath =  
  "/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/  
  unidyn";
```

YOU SHOULD NOT NEED TO EDIT ANYTHING FROM HERE ONWARDS.

Load the package

Append the package path to the system path. Before trying to load the package, ask *Mathematica* to find it. This is a test that we directed *Mathematica* to the correct directory. The output of this command should be the full system path to the UniDyn.m file.

```
$Path = AppendTo[$Path, $PackagePath];  
FindFile["UniDyn`"]
```

```
/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/UniDyn/UniDyn.m
```

Now that we are confident that the path is set correctly, load the package. Setting the global `$VerboseLoad` variable to `True` will print out the help strings for key commands in the package.

```
$VerboseLoad = True;  
Needs["UniDyn`"]
```

You are using the version of NCAlgebra which is found in:

```
/Users/jam99/Downloads/NC
```

You can now use "`<< NCAlgebra``" to load NCAlgebra or "`<< NCGB``" to load NCGB.

```
NCMultiplication.m loaded
```

```
NC1SetCommands.m loaded
```

```
NCInverses.m loaded
```

```
NCTransposes.m loaded
```

```
NCAAdjoints.m loaded
```

```
NCCo.m loaded
```

```
NCRoots.m loaded
```

```
NC2SetCommands.m loaded
```

```
NCCollect.m loaded
```

```
NCSubstitute.m loaded
```

```
NCMonomial.m loaded
```

```
NCSolve.m loaded
```

```
NCTools.m loaded
```

```
NC2SimplifyRational.m loaded
```

```
NC1SimplifyRational.m loaded
```

```
NCSimplifyRational.m loaded
```

```
NCComplex.m loaded
```

```
NCMatMult.m loaded
```

```
NCDiff.m loaded
```

```
NCSchur.m loaded
```

```
NCAlias.m loaded
```

```
Grabs.m loaded
```

```
NCTaylorCoeff.m loaded
```

```
NCConvexity.m and NCGuts.m loaded
```

```
NCTaylorCoeff.m loaded
```

NCTeXForm.m loaded

NCTeX::Using 'open' as PDFViewer.

NCTeX.m loaded

NCMaster.m loaded

NCOutput.m loaded

 NCAlgebra - Version 4.0.6
 Compatible with Mathematica Version 9

Authors:

J. William Helton*
 Mauricio de Oliveira*
 Mark Stankus*
 Robert L. Miller‡

* Math Dept, UCSD
 ‡ General Atomics Corp
 La Jolla, CA 92093

Copyright:

Helton and Miller June 1991
 Helton 2002
 All rights reserved.

The program was written by the authors and by:

David Hurst, Daniel Lamm, Orlando Merino, Robert Obar,
 Henry Pfister, Mike Walker, John Wavrik, Lois Yu,
 J. Camino, J. Griffin, J. Ovall, T. Shaheen, John Shopple.
 The beginnings of the program come from eran@slac.
 Considerable recent help came from Igor Klep.

This program was written with support from
 AFOSR, NSF, ONR, Lab for Math and Statistics at UCSD,
 UCSD Faculty Mentor Program,
 and US Department of Education.
 Primary support in 2010 is from the
 NSF Division of Mathematical Sciences.

If you

- (1) are a user,
 - (2) want to be a user,
 - (3) refer to NCAlgebra in a publication, or
 - (4) have had an interesting experience with NCAlgebra,
- let us know by sending an e-mail message to

ncalg@math.ucsd.edu.

We do not want to restrict access to NCAlgebra, but do
 want to keep track of how it is being used.

For NCAlgebra updates see the web page:

www.math.ucsd.edu/~ncalg

 You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

CreateOperator::usage :

CreateOperator[] is used to batch-define a bunch of operators. Example: CreateOperator[{{lx, ly, lz},{Sx,Sy,Sz}}] will create six operators; each of the operators in the first list is meant to commute with each of the operators in the second list.

CreateScalar::usage :

CreateScalar[list] is used to batch-define a bunch of scalars. The parameter list can be a single scalar or a list of scalars. Example: CreateScalar[{w1,w2}].

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

NCSort::usage : NCSort[list] sorts the operators in list into canonical order.

SortedMult::usage :

SortedMult[list] returns Mult[list\$ordered], where list\$ordered are the elements of list sorted into canonical order.

MultSort::usage :

MultSort[NonCommutativeMultiply[list]] returns NonCommutativeMultiply[list\$ordered], where list\$ordered are the elements of list sorted into canonical order.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

Comm::usage : Comm[a,b] calculates the commutator of two operators.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

SpinSingle\$CreateOperators::usage :

SpinSingle\$CreateOperators[lx,ly,lz,L] creates lx, ly, and lz angular momentum operators and defines their commutation relations. When the total angular momentum $L = 1/2$, additional rules are defined to simplify products of the angular momentum operators. When the total angular momentum L is unspecified, no such simplification rules are defined.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

OscSingle\$CreateOperators::usage :

OscSingle\$CreateOperators[aL,aR] creates a raising operator aR and a lowering operator aL for single harmonic oscillator and defines the operator commutation relations.

You are using the version of NCAlgebra which is found in:

```
/Users/jam99/Downloads/NC
```

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

Evolve::usage :

Evolve[H,t, ρ] represents unitary evolution of the density operator ρ for a time t under the Hamiltonian H.

This function expands according to simplification rules but leaves the evolution unevaluated.

Evolver::usage : Evolver[H,t, $\rho(0)$] calculates $\rho(t) = \text{Exp}[-I H t] \rho(0) \text{Exp}[I H t]$, assuming

that H is time independent, according to the commutation rules followed by $\rho(0)$ and H.

Evolver function with simplification

Free evolution operator

```
SimplifyingEvolver[H_, t_,  $\rho\$0$ _] :=
```

```
(*   H = Hamiltonian operator *)
(*   t = time variable *)
(*  $\rho\$0$  = initial density operator *)
```

```
Evolver[H, t,  $\rho\$0$ ] // Simplify // ExpToTrig // FullSimplify
```

Impliment pulses with simplication

I spins

Evolve the I spins under a short burst of applied on-resonance irradiation, in the delta-function pulse approximation. Specify the total pulse angle and the pulse phase and the PulseI operator evolves the density operator accordingly.

```
Clear[PulseI];
```

```
PulseI[ $\rho\$0$ _,  $\phi$ _,  $\theta$ _] :=
```

```
(*  $\rho\$0$  = initial density operator *)
(*  $\phi$  = pulse phase in radians *)
(*  $\theta$  = pulse angle in radians *)
```

```
 $\rho\$0$  //
```

```
SimplifyingEvolver[ $\omega\$I$  (Cos[ $\phi$ ] Ix + Sin[ $\phi$ ] Iy), t, #] & /.
{t  $\rightarrow$   $\theta / \omega\$I$ }
```

The pulse operator distributes over Plus, Times, and NoncommutativeMultiply. Scalars in front of operators can be pulled out front.

```
PulseI[ρ$0_Plus, φ_, θ_] :=
  Plus @@ (PulseI[#, φ, θ] &) /@ List @@ ρ$0
PulseI[ρ$0_Times, φ_, θ_] :=
  Times @@ (PulseI[#, φ, θ] &) /@ List @@ ρ$0
PulseI[ρ$0_NonCommutativeMultiply, φ_, θ_] :=
  NonCommutativeMultiply @@ (PulseI[#, φ, θ] &) /@ List @@ ρ$0
PulseI[Times[a_?CommutativeQ, ρ$0_], φ_, θ_] :=
  a PulseI[ρ$0, φ, θ]
```

S spins

Define an analogous pulse operator for the S spins ...

```
Clear[PulseS];
PulseS[ρ$0_, φ_, θ_] :=

  (* ρ$0 = initial density operator *)
  (* φ = pulse phase in radians *)
  (* θ = pulse angle in radians*)

  ρ$0 //
  SimplifyingEvolver[ω$S (Cos[φ] Sx + Sin[φ] Sy), t, #] & /.
  {t → θ / ω$S}
```

... with nice distributive and scalar-factoring properties.

```
PulseS[ρ$0_Plus, φ_, θ_] :=
  Plus @@ (PulseS[#, φ, θ] &) /@ List @@ ρ$0
PulseS[ρ$0_Times, φ_, θ_] :=
  Times @@ (PulseS[#, φ, θ] &) /@ List @@ ρ$0
PulseS[ρ$0_NonCommutativeMultiply, φ_, θ_] :=
  NonCommutativeMultiply @@ (PulseS[#, φ, θ] &) /@ List @@ ρ$0
PulseS[Times[a_?CommutativeQ, ρ$0_], φ_, θ_] :=
  a PulseS[ρ$0, φ, θ]
```

Two spins

The assumptions define below are required for *Mathematica* to recognize

$\sqrt{-\Delta^2 - \omega^2} = i \sqrt{\Delta^2 + \omega^2}$ inside an exponential. One of the variables has to be defined to be > 0 and not just ≥ 0 . For *Mathematica* and the Evolve[] function to

recognize unitary rotations, it is crucially important to define all the non-operator variables in your expressions to be scalars.

```
Clear[

  Δ$I,      (* Resonance offset frequency *)
  Δ$S,      (* Resonance offset frequency *)
  J,        (* Spin-spin coupling *)
  Ix, Iy, Iz, (* Spin angular momentum operators *)
  Sx, Sy, Sz, (* Spin angular momentum operators *)
  ρ,        (* Spin density operator *)
  t,        (* Time *)
  t1,       (* Time *)
  t2,       (* Time *)
  a,        (* Curie Law factor *)
  b,        (* Curie Law factor *)
  ρ$0,      (* Initial spin density operator *)
  H         (* Spin Hamiltonian *)

]

CreateScalar[Δ$I, Δ$S, J, t, t1, t2, a, b];

$Assumptions = {Element[Δ$I, Reals], Δ$I ≥ 0,
  Element[Δ$S, Reals], Δ$S ≥ 0, Element[J, Reals], J > 0};

CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz, L = 1 / 2];
SpinSingle$CreateOperators[Sx, Sy, Sz, L = 1 / 2];
SpinSingle$CreateOperators::nocreate : Spin operators already exist.
SpinSingle$CreateOperators::comm : Adding spin commutations relations.
SpinSingle$CreateOperators::simplify : Angular momentum L = 1/2. Adding operator simplification rules.
SpinSingle$CreateOperators::nocreate : Spin operators already exist.
SpinSingle$CreateOperators::comm : Adding spin commutations relations.
SpinSingle$CreateOperators::simplify : Angular momentum L = 1/2. Adding operator simplification rules.
```

Free evolution operator for two coupled spins

Evolve the density operator under the Hamiltonian

$$H = \Delta_I I_z + \Delta_S S_z + J I_z S_z$$

where Δ_I is the I-spin chemical shift, Δ_S is the S-spin chemical shift, and J is the through-bond coupling between the I spin and the S spin. The `Evolve[]` operator cannot evolve the density operator all in one step, so we use the fact that all three terms in the above Hamiltonian commute with each other. This commutation relationship allows us to evolve the density stepwise, by one of the operators at a time.

```
Clear[FreeEvolution];
FreeEvolution[ρ_, t_] :=

(* ρ = density operator *)
(* t = time [s] *)

((ρ // SimplifyingEvolver[Δ$I Iz, t, #] & //
  SimplifyingEvolver[ Δ$S Sz, t, #] & //
  SimplifyingEvolver[ J Iz ** Sz, t, #] &) //
  NCEExpand // MultSort) // NCEExpand
```

The `FreeEvolution` operator distributes over Plus, Times, and NoncommutativeMultiply. Scalars in front of operators can be pulled out front.

```
FreeEvolution[ρ_Plus, t_] :=
  Plus @@ (FreeEvolution[#, t] &) /@ List @@ ρ
FreeEvolution[ρ_Times, t_] :=
  Times @@ (FreeEvolution[#, t] &) /@ List @@ ρ
FreeEvolution[ρ_NonCommutativeMultiply, t_] :=
  NonCommutativeMultiply @@ (FreeEvolution[#, t] &) /@ List @@ ρ
FreeEvolution[Times[a_?CommutativeQ, ρ_], t_] :=
  a FreeEvolution[ρ, t]
```

Load Spin 1/2 matrices

```
<<Matrices--two-spin-half.m
Loaded spin one-half matrices for mIx, mIy, mIz, mSx, mSy, mSz
```

This set of rules will substitute appropriate 4 x 4 matrices for the symbolic spin operators above.

```
SubstituteSpinMatrices = {Ix → mIx,
  Iy → mIy, Iz → mIz, Sx → mSx, Sy → mSy, Sz → mSz} ;
```

Utility functions for one-dimensional data

Make an array of time data

```
TimeArray[NN_, T_] :=
```

```
  (* NN = number of points      *)
  (* T = total acquisition time *)
```

```
  Module[{dt, t},
    dt = T / (NN - 1);
    t = Table[ii*dt, {ii, 0, NN - 1}];
    Return[t]]
```

Print out the time step and the Nyquist frequency

```
TimeArrayReport[t_] :=
```

```
  (* t = List of time points *)
```

```
  Module[{dt},
    dt = t[[2]] - t[[1]];
    Print["The time step is ", dt, " s"];
    Print["The Nyquist frequency is ", 1 / (2*dt), " Hz"]]
```

Calculate the digital Fourier transform of a one-dimensional time-domain signal and plot it.

```
DFFT[signal_, t_, query_: True] :=
```

```
  (* signal = A function that computes the signal *)
  (*      t = A list of time points *)
  (*  query = True or False *)
  (*      True to plot the real part,
imaginary part, and magnitude of the FT *)
  (*      False to plot only
the real part of the FT *)
```

```
  Module[{NN, (* number of data points *)
    T, (* total acquisition time;
assumes data starts at zero *)
    f (* array of frequency points *)
  },
```

```
    NN = Dimensions[signal /@ t][[1]];
    FFTS = RotateRight[Fourier[signal /@ t], NN / 2];
```

```

T = t[[-1]];
f = Table[jj/T, {jj, -NN/2, NN/2 - 1}];

p1 = ListLinePlot[
  {Transpose[{t, Re[signal/@t]}],
   Transpose[{t, Im[signal/@t]}]},
  PlotRange → {-1, 1},
  AxesLabel → {"time [s]", "signal"}];

If[query == True,

  p2 = ListLinePlot[
    {Transpose[{f, Re[FFTS]}],
     Transpose[{f, Im[FFTS]}],
     Transpose[{f, Abs[FFTS]}] },
    PlotRange → All,
    AxesLabel → {"freq [Hz]", "DFT{signal}"}],

  p2 = ListLinePlot[
    Transpose[{f, Re[FFTS]}],
    PlotRange → All,
    AxesLabel → {"freq [Hz]", "DFT{signal}"}]
];

Show[GraphicsGrid[{{p1, p2}}]]
]

```

Calculate the digital Fourier transform of a one-dimensional time-domain signal and plot it.

DFFT\$all[signal_, t_] :=

```

(* signal = A function that computes the signal *)
(*      t = A list of time points *)
(* query = True or False *)
(*      True to plot the real part,
imaginary part, and magnitude of the FT *)
(*      False to plot only
the real part of the FT *)

```

```

Module[{NN, (* number of data points *)
      T, (* total acquisition time;
      assumes data starts at zero *)
      f (* array of frequency points *)
},

NN = Dimensions[signal /@ t][[1]];
FFTS = RotateRight[Fourier[signal /@ t], NN/2];

T = t[[-1]];
f = Table[jj/T, {jj, -NN/2, NN/2 - 1}];

p1 = ListLinePlot[
  Transpose[{f, Re[FFTS]}],
  PlotRange -> All,
  AxesLabel -> {"freq [Hz]", "Re[DFT{signal}"]}];

p2 = ListLinePlot[
  Transpose[{f, Im[FFTS]}],
  PlotRange -> All,
  AxesLabel -> {"freq [Hz]", "Im[DFT{signal}"]}];

p3 = ListLinePlot[
  Transpose[{f, Abs[FFTS]}],
  PlotRange -> All,
  AxesLabel -> {"freq [Hz]", "Abs[DFT{signal}"]}];

Show[GraphicsGrid[Transpose[{{p1, p2, p3}}]]]
]

```

Free-evolution signal

Edit the chemical shifts and the J-couplings here

```

HamiltonianParameters$coupled =

{ $\Delta I \rightarrow 2 \pi * 100$ , (* I-spin chemical shift [radians] *)
 $\Delta S \rightarrow 2 \pi * 300$ ,
(* S-spin chemical shift [radians] *)
 $J \rightarrow 2 \pi * 50$ ,      (* scalar coupling [radians] *)
T2 -> 0.3              (* spin dephasing time [s] *)
};

```

Make a copy of the parameters with the scalar coupling set to zero, e.g., with the two spins uncoupled. Useful for testing.

```

HamiltonianParameters$uncoupled =
  HamiltonianParameters$coupled;
HamiltonianParameters$uncoupled[[3]] =  $J \rightarrow 0$ ;

```

Evolve the initial density operator $I_x + S_x$ under the coupled-spin Hamiltonian. Add an exponential decay so we get a reasonable looking signal. Calculate the signal corresponding to observation of the complex signal $(I_x + S_x) + i(I_y + S_{xy})$ using the matrix form of the operators.

```

 $\rho$  = FreeEvolution[Ix + Sx, t] Exp[-t / T2];
Signal[t_] = Tr[ $\rho$  . ((Ix + Sx) + I (Iy + Sy))] /.
  SubstituteSpinMatrices /.
  HamiltonianParameters$coupled;

```

Create an array of 1024 time points.

```

t$List = TimeArray[2^10, 1.0];
t$List // TimeArrayReport


```

The time step is 0.000977517 s

The Nyquist frequency is 511.5 Hz


Calculate the time-domain signal. Fourier transform it to reveal the spectrum of the two coupled spins.



DFFT[Signal, t\$List, True]



Let's plot the real, imaginary, and absolute value signals separately. Notice how broad the

DFFT\$all[Signal, t\$List]






Recalculate the signal with the spin-spin coupling dialed to zero. Note how the


doublet pattern in the spectrum vanishes.

```
 $\rho$  = FreeEvolution[(Ix + Sx) , t] Exp[-t / T2];
Signal[t_] = Tr[ $\rho$  . ((Ix + Sx) + I (Iy + Sy)) ] /.
  SubstituteSpinMatrices /.
  HamiltonianParameters$uncoupled;
DFFT[Signal, t$List, False]
```




Evolve just I_x . We see just one of the doublets.

```
 $\rho$  = FreeEvolution[Ix , t] Exp[-t / T2];
Signal[t_] = Tr[ $\rho$  . ((Ix + Sx) + I (Iy + Sy)) ] /.
  SubstituteSpinMatrices /.
  HamiltonianParameters$coupled;
DFFT[Signal, t$List, False]
```



Evolve just S_x . See just the other doublet.

```
ρ = FreeEvolution[Sx , t] Exp[-t / T2];
Signal[t_] = Tr[ρ . ((Ix + Sx) + I (Iy + Sy)) ] /.
  SubstituteSpinMatrices /.
  HamiltonianParameters$coupled;
DFFT[Signal, t$List, False]
```



INEPT polarization transfer

Here INEPT is an acronym that stands for *insensitive nuclei enhanced by polarization transfer*. Thinking through the INEPT experiment helps you understand how two-dimensional spectroscopy works.

The INEPT experiment shows how a J coupling can be harnessed to transfer spin polarization from a nucleus with a large magnetic moment -- ^1H , the I-spin here -- to a nucleus with a small magnetic moment -- ^{13}C , the S-spin here. At thermal equilibrium the initial density operator is

$$\rho_{\text{equil}} = a I_z + b S_z$$

with a and b the Curie factors for each of the spins. We worked through these Curie-law factors in detail on a homework. To discuss the INEPT experiment it is enough to know that the factor b is about 1/16 the size of the a factor:

$$\text{SubstitutePrefactors} = \left\{ a \rightarrow 1, \quad b \rightarrow \left(\frac{1}{4} \right)^2 \right\};$$

In other words, the thermal-equilibrium ^{13}C signal is about 1/16 as large as the thermal-equilibrium ^1H signal in a magnetic resonance experiment.

The usual ^{13}C spectrum

Let's calculate the usual ^{13}C signal. Deliver a $\pi/2$ pulse to the ^{13}C spins and then evolve the spins under the Hamiltonian

$$H = \Delta_I I_z + \Delta_S S_z + J I_z S_z$$

For simplicity we will work on resonance where $\Delta_I \rightarrow 0$ and $\Delta_S \rightarrow 0$.

```

ρ$0 = a Iz + b Sz // PulseS[#, π/2, π/2] &;
ρ$1 =
  (ρ$0 // FreeEvolution[#, t] & // NCEExpand) /. {Δ$I → 0, Δ$S → 0}
a Iz + b Sx Cos[ $\frac{J t}{2}$ ] + 2 b Iz ** Sy Sin[ $\frac{J t}{2}$ ]

```


We detect the complex magnetization and Fourier transform it to obtain the ^{13}C spectrum.

```

Signal[t_] =
  Tr[ρ$1.(Sx + I Sy)] Exp[-t/T2] /. SubstituteSpinMatrices /.
  HamiltonianParameters$coupled /. SubstitutePrefactors
 $\frac{1}{16} e^{-3.33333 t} \cos[50 \pi t]$ 

```

```
DFFT[Signal, t$List, False]
```



The spectrum shows a symmetric doublet, due the J-coupling, that is centered

near zero frequency. The height of each peak in the spectrum is approximately ~ 0.28 .

The ^{13}C spectrum following polarization transfer from ^1H

The INEPT experiment consists of a string of pulses applied to the I-spins and the S-spins.

$$\begin{aligned} \text{I spins: } & (\pi/2)_y - \tau - (\pi)_x - \tau - (\pi/2)_x \\ \text{S spins: } & (\pi)_x - \tau - (\pi/2)_y - \text{detect} \end{aligned}$$

with the time delay set to $\tau = 2\pi / (4J)$, with J the scalar coupling. Now let's calculate the usual ^{13}C signal following the INEPT polarization-transfer sequence. For simplicity we will again work on resonance where $\Delta_I \rightarrow 0$ and $\Delta_S \rightarrow 0$. in the calculation below I print out the density operator at each step in the evolution.

```

ρ$0 = a Iz + b Sz // PulseI[#, π/2, π/2] &
ρ$1 = (ρ$0 // FreeEvolution[#, t] & // NCExpand) /. {Δ$I → 0}
ρ$2 = ρ$1 // PulseI[#, 0, π] & // PulseS[#, 0, π] &
ρ$3 = (ρ$2 // FreeEvolution[#, t] & // NCExpand) /. {Δ$I → 0}
ρ$4 = ρ$3 // PulseI[#, 0, π/2] & // PulseS[#, π/2, π/2] &
a Ix + b Sz

```

$$\begin{aligned} & b Sz + a Ix \cos\left[\frac{Jt}{2}\right] + 2 a Iy ** Sz \sin\left[\frac{Jt}{2}\right] \\ & - b Sz + a Ix \cos\left[\frac{Jt}{2}\right] + 2 a Iy ** Sz \sin\left[\frac{Jt}{2}\right] \\ & - b Sz + a Ix \cos\left[\frac{Jt}{2}\right]^2 + 4 a \cos\left[\frac{Jt}{2}\right] Iy ** Sz \sin\left[\frac{Jt}{2}\right] - a Ix \sin\left[\frac{Jt}{2}\right]^2 \\ & - b Sx + a Ix \cos\left[\frac{Jt}{2}\right]^2 + 4 a \cos\left[\frac{Jt}{2}\right] Iz ** Sx \sin\left[\frac{Jt}{2}\right] - a Ix \sin\left[\frac{Jt}{2}\right]^2 \end{aligned}$$

The signal in the “regular” experiment arose from the term

$$b Sx \cos\left[\frac{Jt}{2}\right]$$

in the density operator. After the series of pulses in the INEPT experiment we now have a term in the density operator given by

$$4 a \cos\left[\frac{Jt}{2}\right] Iz ** Sx \sin\left[\frac{Jt}{2}\right] \rightarrow 2 a Iz ** Sx$$

where the term following the arrow is what we get by setting the delay time to $2\pi/(4J)$. This term arose from the *initial I-spin signal* evolving under the spin-spin coupling to give a *mixed I-spin & S-spin signal*. Further evolution under the J coupling Hamiltonian will give rise a *pure S-spin signal*.

```
ρ$5 = ρ$4 /. {t → 2 π / (4 J)} // FreeEvolution[#, t] & //
NCEExpand (* Why is an extra NCEExpand needed here? *)

-b Sx Cos[ $\frac{J t}{2}$ ] Cos[t Δ$S] + 2 a Cos[ $\frac{J t}{2}$ ] Cos[t Δ$S] Iz ** Sx +

a Sy Cos[t Δ$S] Sin[ $\frac{J t}{2}$ ] - 2 b Cos[t Δ$S] Iz ** Sy Sin[ $\frac{J t}{2}$ ] - b Sy Cos[ $\frac{J t}{2}$ ] Sin[t Δ$S] +


2 a Cos[ $\frac{J t}{2}$ ] Iz ** Sy Sin[t Δ$S] - a Sx Sin[ $\frac{J t}{2}$ ] Sin[t Δ$S] + 2 b Iz ** Sx Sin[ $\frac{J t}{2}$ ] Sin[t Δ$S]
```

Remarkably, **the ^{13}C signal is now approximately 16x larger**. This signal increase corresponds to a $16^2 = 256$ -fold savings in signal averaging time.

```
Signal[t_] = Tr[ρ$5.(Sx + I Sy)] Exp[-t / T2] /. {Δ$I → 0, Δ$S → 0} /.
SubstituteSpinMatrices /.
HamiltonianParameters$coupled /. SubstitutePrefactors
```

```
e-3.33333 t  $\left( -\frac{1}{16} \cos[50 \pi t] + i \sin[50 \pi t] \right)$ 
```

```
DFFT[Signal, t$List, False]
```



Utility functions for two-dimensional data

Function to take the 2D Fourier Transform

```

DFFT2[signal_, t1_, t2_, query_:False] :=

(* signal = A function that computes the signal *)
(*      t1 = A list of time points *)
(*      t2 = A list of time points *)
(*  query = True or False,
whether to print out diagnostic messages *)

Module[{t1$NN,
  (* number of time points in the indirect, t1 dimension *)
  t2$NN, (* number of time points in the direct,
t2 dimension *)
  f1,    (* the frequency axis
in the indirect dimension *)
  f2     (* the frequency axis in
the indirect dimension *)
},

(* Too fancy: signal$List = Outer[N[signal[#1,#2]]&,t1,t2]; *)

t1$NN = Dimensions[t1][[1]];
t2$NN = Dimensions[t2][[1]];

signal$List = Table[signal[t1[[i$1]], t2[[i$2]]] // N,
  {i$2, 1, t2$NN}, {i$1, 1, t1$NN}];

If[query == True,
  Print["Number of t1 points = ", t1$NN];
  Print["Number of t2 points = ", t2$NN];
  Print["Dimensions of signal$List = ",
    Dimensions[signal$List]];
];

signal$List$FT =
  RotateLeft[Fourier[signal$List], {t2$NN/2, t1$NN/2}];

f1 = Table[jj/t1[[-1]], {jj, -t1$NN/2, t1$NN/2-1}];
f2 = Table[jj/t2[[-1]], {jj, -t2$NN/2, t2$NN/2-1}];

SetOptions[ListPlot3D,
  PlotRange → All,
  DataRange → {{f1[[1]], f1[[t1$NN]]}, {f2[[1]], f2[[t2$NN]]}},
  PlotLabel → "2D spectrum\n\n",
  Mesh → False];

```

```

p1 = ListPlot3D[Re[signal$List$FT],
  AxesLabel → {"f1 [Hz]", "f2 [Hz]", "Re[FT[signal(t1,t2)]]"},
  ViewPoint → {1.561`, -2.881`, 0.845`}];

p2 = ListPlot3D[Im[signal$List$FT],
  AxesLabel → {"f1 [Hz]", "f2 [Hz]", "Im[FT[signal(t1,t2)]]"},
  ViewPoint → {1.561`, -2.881`, 0.845`}];

p3 = ListPlot3D[Abs[signal$List$FT],
  AxesLabel → {"f1 [Hz]", "f2 [Hz]", "Abs[FT[signal(t1,t2)]]"},
  ViewPoint → {1.561`, -2.881`, 0.845`}];

Return[{p1, p2, p3}]
]

```

Test case #I

```

t1$List = TimeArray[2^6, 0.25];
t2$List = TimeArray[2^7, 0.50];

t1$List // TimeArrayReport
t2$List // TimeArrayReport

The time step is 0.00396825 s
The Nyquist frequency is 126. Hz

The time step is 0.00393701 s
The Nyquist frequency is 127. Hz

Signal2D[t1_, t2_] = Exp[-I 2 π 100 t1]
  Exp[I 2 π 50 t2] Exp[-t1 / 0.08] Exp[-t2 / 0.15] + 0.05;
{p1, p2, p3} = DFFT2[Signal2D, t1$List, t2$List, True];

p3


```

Number of t1 points = 64

Number of t2 points = 128

Dimensions of signal\$List = {128, 64}

2D spectrum



Test Case #2

```
t1$List = TimeArray[2^9, 0.25];
```

```
t2$List = TimeArray[2^6, 0.50];
```

```
t1$List // TimeArrayReport
```

```
t2$List // TimeArrayReport
```

```
Signal2D[t1_, t2_] = Exp[-I 2 π 300 t1]
```

```
Exp[-I 2 π 30 t2] Exp[-t1 / 0.05] Exp[-t2 / 0.10] + 0.05;
```

```
{p1, p2, p3} = DFFT2[Signal2D, t1$List, t2$List, True];
```


```
p3
```

```

The time step is 0.000489237 s
The Nyquist frequency is 1022. Hz
The time step is 0.00793651 s
The Nyquist frequency is 63. Hz
Number of t1 points = 512
Number of t2 points = 64
Dimensions of signal$List = {64, 512}

```

2D spectrum



Heteronuclear COSY signal

This is the signal that we worked out in class

Set the couplings and number of points

Edit the chemical shifts and the J-couplings here

```

HamiltonianParameters$coupled =

{Δ$I → 2 π * 100, (* I-spin chemical shift [radians] *)
 Δ$S → 2 π * 150, (* S-spin chemical shift [radians] *)
 J → 2 π * 40,    (* scalar coupling [radians] *)
 T2 -> 0.10      (* spin dephasing time [s] *)
};

```

Edit the number of points at the total acquisition time here

```

t1$List = TimeArray[2^7, 0.25];
t2$List = TimeArray[2^7, 0.25];

```

Make a copy of the parameters with the scalar coupling set to zero, e.g., with the two spins uncoupled. Useful for testing.


```

HamiltonianParameters$uncoupled =
  HamiltonianParameters$coupled;
HamiltonianParameters$uncoupled[[3]] = J → 0;

```

Some checks

Check out the Nyquist frequencies

```

t1$List // TimeArrayReport
t2$List // TimeArrayReport

```

The time step is 0.0019685 s

The Nyquist frequency is 254. Hz

The time step is 0.0019685 s

The Nyquist frequency is 254. Hz

Evolve the density operator

Compute evolution. This computation takes a couple of seconds.

```
ρ$0 = Iz + Sz;
```

```
(* Initial (π/2)y pulse*)
```

```

ρ$1$A = ρ$0 // PulseS[#, π/2, π/2] &;
ρ$2$A = ρ$1$A // FreeEvolution[#, t1] &;
ρ$3$A = ρ$2$A // PulseS[#, π/2, π/2] & // PulseI[#, π/2, π/2] &;
ρ$4$A = ρ$3$A // FreeEvolution[#, t2] & // NCExpand // MultSort;

```

```
(* Initial (π/2)x pulse*)
```

```

ρ$1$B = ρ$0 // PulseS[#, 0, π/2] &;
ρ$2$B = ρ$1$B // FreeEvolution[#, t1] &;
ρ$3$B = ρ$2$B // PulseS[#, π/2, π/2] & // PulseI[#, π/2, π/2] &;
ρ$4$B = ρ$3$B // FreeEvolution[#, t2] & // NCExpand // MultSort;

```

Examine the two signals and combine them

```
(Tr[ρ$4$A.(Ix + I Iy)] /. SubstituteSpinMatrices // Simplify)
```

$$(\cos[t_2 \Delta S I] + i \sin[t_2 \Delta S I]) \left(\cos\left[\frac{J t_2}{2}\right] + i \sin\left[\frac{J t_1}{2}\right] \sin\left[\frac{J t_2}{2}\right] \sin[t_1 \Delta S S] \right)$$

```
(Tr[ρ$4$B.(Ix + I Iy)] /. SubstituteSpinMatrices // Simplify)
```

$$\left(\cos\left[\frac{J t_2}{2}\right] - i \cos[t_1 \Delta S S] \sin\left[\frac{J t_1}{2}\right] \sin\left[\frac{J t_2}{2}\right] \right) (\cos[t_2 \Delta S I] + i \sin[t_2 \Delta S I])$$

```

ρ$4 = -ρ$4$A + I ρ$4$B;
(Tr[ρ$4.(Ix + I Iy)] /. SubstituteSpinMatrices // Simplify)
(Cos[t2 Δ$I] + i Sin[t2 Δ$I])

$$\left( (-1 + i) \cos\left[\frac{J t_2}{2}\right] + \sin\left[\frac{J t_1}{2}\right] \sin\left[\frac{J t_2}{2}\right] (\cos[t_1 \Delta S] - i \sin[t_1 \Delta S]) \right)$$


```

Compute the signal and apply the 2D FT

```

Signal2D[t1_, t2_] =

Tr[ρ$4.(Ix + I Iy)] Exp[-t1/T2] Exp[-t2/T2] + 0.025 /.
SubstituteSpinMatrices /.
HamiltonianParameters$coupled // Simplify // N;


```


Fourier transform the signal. This takes a minute or so.

```
{p1, p2, p3} = DFFFT2[Signal2D, t1$List, t2$List];
```

p1

p3





Homonuclear COSY signal

This is the signal you get if you pulse and detect *both* nuclei at the same time.

Set the couplings and number of points

Edit the chemical shifts and the J-couplings here

HamiltonianParameters\$coupled =

```
{Δ$I → 2 π * 50, (* I-spin chemical shift [radians] *)
 Δ$S → 2 π * 150, (* S-spin chemical shift [radians] *)
 J → 2 π * 23,    (* scalar coupling [radians] *)
 T2 -> 0.10      (* spin dephasing time [s] *)
};
```

Edit the number of points at the total acquisition time here

```
t1$List = TimeArray[2^7, 0.25];
t2$List = TimeArray[2^7, 0.25];
```

Make a copy of the parameters with the scalar coupling set to zero, e.g., with the two spins uncoupled. Useful for testing.

```
HamiltonianParameters$uncoupled =
  HamiltonianParameters$coupled;
HamiltonianParameters$uncoupled[[3]] = J → 0;
```

Some checks

Check out the Nyquist frequencies

```
t1$List // TimeArrayReport
t2$List // TimeArrayReport
```

The time step is 0.0019685 s

The Nyquist frequency is 254. Hz

The time step is 0.0019685 s

The Nyquist frequency is 254. Hz

Evolve the density operator

Compute evolution. This computation can take up to a minute.

```
 $\rho\$0 = I_z + S_z;$ 
```

```
(* Initial  $(\pi/2)$ y pulses*)
```

```
 $\rho\$1\$A = \rho\$0$  // PulseS[#,  $\pi/2$ ,  $\pi/2$ ] & // PulseI[#,  $\pi/2$ ,  $\pi/2$ ] & //  
  NCEExpand // MultSort;  
 $\rho\$2\$A = \rho\$1\$A$  // FreeEvolution[#, t1] & // NCEExpand // MultSort;  
 $\rho\$3\$A = \rho\$2\$A$  // PulseS[#,  $\pi/2$ ,  $\pi/2$ ] & // PulseI[#,  $\pi/2$ ,  $\pi/2$ ] & //  
  NCEExpand // MultSort;  
 $\rho\$4\$A = \rho\$3\$A$  // FreeEvolution[#, t2] & // NCEExpand // MultSort;
```

```
(* Initial  $(\pi/2)$ x pulses*)
```

```
 $\rho\$1\$B =$   
   $\rho\$0$  // PulseS[#, 0,  $\pi/2$ ] & // PulseI[#, 0,  $\pi/2$ ] & // NCEExpand //  
    MultSort;  
 $\rho\$2\$B = \rho\$1\$B$  // FreeEvolution[#, t1] & // NCEExpand // MultSort;  
 $\rho\$3\$B = \rho\$2\$B$  // PulseS[#,  $\pi/2$ ,  $\pi/2$ ] & // PulseI[#,  $\pi/2$ ,  $\pi/2$ ] & //  
  NCEExpand // MultSort;  
 $\rho\$4\$B = \rho\$3\$B$  // FreeEvolution[#, t2] & // NCEExpand // MultSort;
```

Examine the two signals and combine them

```
(Tr[ $\rho\$4\$A \cdot ((I_x + S_x) + I(I_y + S_y))$ ] /.  
  SubstituteSpinMatrices // Simplify)
```

$$\begin{aligned} & i \left(\cos\left[\frac{J t_1}{2}\right] \cos\left[\frac{J t_2}{2}\right] (\cos[t_2 \Delta I] \sin[t_1 \Delta I] + \right. \\ & \quad i \sin[t_1 \Delta I] \sin[t_2 \Delta I] + \sin[t_1 \Delta S] (\cos[t_2 \Delta S] + i \sin[t_2 \Delta S])) + \\ & \quad \sin\left[\frac{J t_1}{2}\right] \sin\left[\frac{J t_2}{2}\right] (\cos[t_2 \Delta S] \sin[t_1 \Delta I] + \cos[t_2 \Delta I] \sin[t_1 \Delta S] + \\ & \quad \left. i (\sin[t_2 \Delta I] \sin[t_1 \Delta S] + \sin[t_1 \Delta I] \sin[t_2 \Delta S])) \right) \end{aligned}$$

```
(Tr[ $\rho\$4\$B \cdot ((I_x + S_x) + I(I_y + S_y))$ ] /.  
  SubstituteSpinMatrices // Simplify)
```

$$\begin{aligned} & \sin\left[\frac{J t_1}{2}\right] \sin\left[\frac{J t_2}{2}\right] (-i \cos[t_2 \Delta I] \cos[t_1 \Delta S] + \cos[t_1 \Delta S] \sin[t_2 \Delta I] + \\ & \quad \cos[t_1 \Delta I] (-i \cos[t_2 \Delta S] + \sin[t_2 \Delta S])) + \cos\left[\frac{J t_1}{2}\right] \cos\left[\frac{J t_2}{2}\right] \\ & \quad (\cos[t_1 \Delta I] (-i \cos[t_2 \Delta I] + \sin[t_2 \Delta I]) + \cos[t_1 \Delta S] (-i \cos[t_2 \Delta S] + \sin[t_2 \Delta S])) \end{aligned}$$

```

ρ$4 = -ρ$4$A + I ρ$4$B;
(Tr[ρ$4 . ((Ix + Sx) + I (Iy + Sy))]) /.
  SubstituteSpinMatrices // Simplify)

Sin[ $\frac{J t_1}{2}$ ] Sin[ $\frac{J t_2}{2}$ ] (-i Cos[t2 Δ$S] Sin[t1 Δ$I] + i Cos[t1 Δ$S] Sin[t2 Δ$I] +
  Cos[t2 Δ$I] (Cos[t1 Δ$S] - i Sin[t1 Δ$S]) + Sin[t2 Δ$I] Sin[t1 Δ$S] +
  Cos[t1 Δ$I] (Cos[t2 Δ$S] + i Sin[t2 Δ$S]) + Sin[t1 Δ$I] Sin[t2 Δ$S]) +
  2 Cos[ $\frac{J t_1}{2}$ ] Cos[ $\frac{J t_2}{2}$ ] Cos[ $\frac{1}{2} (t_1 - t_2) (\Delta I - \Delta S)$ ]
  (Cos[ $\frac{1}{2} (t_1 - t_2) (\Delta I + \Delta S)$ ] - i Sin[ $\frac{1}{2} (t_1 - t_2) (\Delta I + \Delta S)$ ])

```

Compute the signal and apply the 2D FT

```

Signal2D[t1_, t2_] =


  Tr[ρ$4 . ((Ix + Sx) + I (Iy + Sy))]
  Exp[-t1 / T2] Exp[-t2 / T2] + 0.025 /.
  SubstituteSpinMatrices /.
  HamiltonianParameters$coupled // Simplify;

```

Fourier transform the signal. This takes a couple of seconds.

```
{p1, p2, p3} = DFFT2[Signal2D, t1$List, t2$List];
```

p3



Scratch